LieAlgebra Lessons - Maple Programming Help

Home : Support : Online Help : Mathematics : DifferentialGeometry : Lessons and Tutorials : LieAlgebras : DifferentialGeometry/LessonsAndTutorials/LieAlgebras/LieAlgebraData

LieAlgebra Lessons

Lesson 1: Creating Lie Algebras

Overview

Lie algebras arise in many different ways.  In the LieAlgebras package, the command LieAlgebraData is used to convert the different ways of defining a Lie algebra into a common format, called a Lie algebra data structure.  Once this data structure has been created, the Lie algebra can be initialized with the DGsetup program.  This initialization process stores information about the Lie algebra in memory so that calculations can be performed with this Lie algebra.  Many Lie algebras can be initialized within a single Maple session.

In this lesson, you will learn to create a Lie algebra in DifferentialGeometry

 – from its structure equations.
 – from a matrix algebra.
 – from a Lie algebra of vector fields.
 – from a subalgebra of an existing Lie algebra.
 – from a DifferentialGeometry library.

You will also learn to do the following:

 – Initialize a Lie algebra and specify the labels for a basis of the Lie algebra.
 – View the multiplication table for a Lie algebra.

Construct a Lie algebra from its structure equations

 > with(DifferentialGeometry): with(LieAlgebras):

To construct a Lie algebra from its structure equations, list the structure equations for a Lie algebra you wish to analyze using a set of unassigned names, in this example, [x1, x2, x3].  Only the non-zero bracket relations need be given.

 > StructureEquations := [[x1, x2] = x3, [x3, x1] = - x2, [x2, x3] = x1];
 ${\mathrm{StructureEquations}}{≔}\left[\left[{\mathrm{x1}}{,}{\mathrm{x2}}\right]{=}{\mathrm{x3}}{,}\left[{\mathrm{x3}}{,}{\mathrm{x1}}\right]{=}{-}{\mathrm{x2}}{,}\left[{\mathrm{x2}}{,}{\mathrm{x3}}\right]{=}{\mathrm{x1}}\right]$ (2.1)

Now pass this information to LieAlgebraData.

 > L := LieAlgebraData(StructureEquations, [x1, x2, x3], Alg1);
 ${L}{≔}\left[\left[{\mathrm{e1}}{,}{\mathrm{e2}}\right]{=}{\mathrm{e3}}{,}\left[{\mathrm{e1}}{,}{\mathrm{e3}}\right]{=}{\mathrm{e2}}{,}\left[{\mathrm{e2}}{,}{\mathrm{e3}}\right]{=}{\mathrm{e1}}\right]$ (2.2)

Take a quick look at the internal representation of the Lie algebra data structure L. This data structure encodes the name of the algebra, Alg1, the dimension 3, and the non-zero structure constants.  For example, the list [[1, 2, 3], 1] indicates that the Lie bracket of e1 and e2 has a component of e3 and the coefficient of that component is 1.

 > lprint(L);
 _DG([["LieAlgebra", Alg1, [3]], [[[1, 2, 3], 1], [[1, 3, 2], 1], [[2, 3, 1], 1]]])

Now initialize the Lie algebra Alg1.  Use the MultiplicationTable command to view the structure equations of the Lie algebra in various formats.

 > DGsetup(L);
 ${\mathrm{Lie algebra: Alg1}}$ (2.3)
 Alg1 > MultiplicationTable("LieTable");
 $\left[\begin{array}{ccccc}{}& {\mathrm{|}}& {\mathrm{e1}}& {\mathrm{e2}}& {\mathrm{e3}}\\ {}& {\mathrm{---}}& {\mathrm{----}}& {\mathrm{----}}& {\mathrm{----}}\\ {\mathrm{e1}}& {\mathrm{|}}& {0}& {\mathrm{e3}}& {\mathrm{e2}}\\ {\mathrm{e2}}& {\mathrm{|}}& {-}{\mathrm{e3}}& {0}& {\mathrm{e1}}\\ {\mathrm{e3}}& {\mathrm{|}}& {-}{\mathrm{e2}}& {-}{\mathrm{e1}}& {0}\end{array}\right]$ (2.4)
 Alg1 > MultiplicationTable("ExteriorDerivative");
 ${d}{}\left({\mathrm{θ1}}\right){=}{-}{\mathrm{θ2}}{}{\mathrm{^}}{}{\mathrm{θ3}}$ ${d}{}\left({\mathrm{θ2}}\right){=}{-}{\mathrm{θ1}}{}{\mathrm{^}}{}{\mathrm{θ3}}$ ${d}{}\left({\mathrm{θ3}}\right){=}{-}{\mathrm{θ1}}{}{\mathrm{^}}{}{\mathrm{θ2}}$ (2.5)

Construct an abstract Lie algebra from a matrix Lie algebra

A vector space M of matrices define a matrix Lie algebra if it is closed with respect to the commutator bracket [a,b] = a*b - b*a, where a and b are matrices in M.  The command LieAlgebraData will compute the commutators of a given list M of matrices, check that each commutator is a linear combination of the elements of M, and return the resulting Lie algebra data structure.

 Alg1 > with(DifferentialGeometry): with(LieAlgebras):

Define a basis for the vector space of 2x2 trace-free matrices.

 Alg1 > M := [Matrix([[1, 0], [0, -1]]), Matrix([[0, 1], [0, 0]]), Matrix([[0, 0], [1, 0]])];
 ${M}{≔}\left[\left[\begin{array}{rr}{1}& {0}\\ {0}& {-}{1}\end{array}\right]{,}\left[\begin{array}{rr}{0}& {1}\\ {0}& {0}\end{array}\right]{,}\left[\begin{array}{rr}{0}& {0}\\ {1}& {0}\end{array}\right]\right]$ (3.1)

Check that the 2x2 trace-free matrices form a Lie algebra and calculate the structure equations.

 $\left[\left[{\mathrm{e1}}{,}{\mathrm{e2}}\right]{=}{2}{}{\mathrm{e2}}{,}\left[{\mathrm{e1}}{,}{\mathrm{e3}}\right]{=}{-}{2}{}{\mathrm{e3}}{,}\left[{\mathrm{e2}}{,}{\mathrm{e3}}\right]{=}{\mathrm{e1}}\right]$ (3.2)

To perform subsequent computations with the Lie algebra Alg2, pass the Lie algebra data structure L to DGsetup.

Construct an abstract Lie algebra from a Lie algebra of vector fields

A list Gamma of independent (over the real numbers) vector fields defines a Lie algebra if for each pair of vectors X, Y in Gamma, the Lie bracket [X,Y] is a constant coefficient linear combination of vector fields in Gamma.  The command LieAlgebraData will compute the Lie brackets of a given list Gamma of vector fields, check that each bracket is a (constant) linear combination of the elements of Gamma, and return the resulting Lie algebra data structure.

 Alg1 > with(DifferentialGeometry): with(LieAlgebras):

Initialize a 2 dimensional manifold N.

 > DGsetup([x, y], N);
 ${\mathrm{frame name: N}}$ (4.1)

Define a list of vector fields.

 N > Gamma := evalDG([D_x, D_y, x*D_x, y*D_x, x*D_y, y*D_y]);
 ${\mathrm{Γ}}{≔}\left[{\mathrm{D_x}}{,}{\mathrm{D_y}}{,}{x}{}{\mathrm{D_x}}{,}{y}{}{\mathrm{D_x}}{,}{x}{}{\mathrm{D_y}}{,}{y}{}{\mathrm{D_y}}\right]$ (4.2)

Check that the list of vector fields Gamma defines a Lie algebra and calculate the structure equations.

 N > L := LieAlgebraData(Gamma, Alg3);
 $\left[\left[{\mathrm{e1}}{,}{\mathrm{e3}}\right]{=}{\mathrm{e1}}{,}\left[{\mathrm{e1}}{,}{\mathrm{e5}}\right]{=}{\mathrm{e2}}{,}\left[{\mathrm{e2}}{,}{\mathrm{e4}}\right]{=}{\mathrm{e1}}{,}\left[{\mathrm{e2}}{,}{\mathrm{e6}}\right]{=}{\mathrm{e2}}{,}\left[{\mathrm{e3}}{,}{\mathrm{e4}}\right]{=}{-}{\mathrm{e4}}{,}\left[{\mathrm{e3}}{,}{\mathrm{e5}}\right]{=}{\mathrm{e5}}{,}\left[{\mathrm{e4}}{,}{\mathrm{e5}}\right]{=}{-}{\mathrm{e3}}{+}{\mathrm{e6}}{,}\left[{\mathrm{e4}}{,}{\mathrm{e6}}\right]{=}{-}{\mathrm{e4}}{,}\left[{\mathrm{e5}}{,}{\mathrm{e6}}\right]{=}{\mathrm{e5}}\right]$ (4.3)

To perform subsequent computations with the Lie algebra Alg3, pass the Lie algebra data structure L to DGsetup.

Construct a Lie algebra by a change of basis in an existing Lie algebra

 > with(DifferentialGeometry): with(LieAlgebras):

Define a Lie algebra algebra data structure and initialize the Lie algebra.

 > L1 := _DG([["LieAlgebra", Alg1, [4]], [[[2, 3, 1], 1], [[1, 4, 1], 2], [[2, 4, 2], 1], [[3, 4, 2], 1], [[3, 4, 3], 1]]]);
 ${\mathrm{L1}}{≔}\left[\left[{\mathrm{e1}}{,}{\mathrm{e4}}\right]{=}{2}{}{\mathrm{e1}}{,}\left[{\mathrm{e2}}{,}{\mathrm{e3}}\right]{=}{\mathrm{e1}}{,}\left[{\mathrm{e2}}{,}{\mathrm{e4}}\right]{=}{\mathrm{e2}}{,}\left[{\mathrm{e3}}{,}{\mathrm{e4}}\right]{=}{\mathrm{e2}}{+}{\mathrm{e3}}\right]$ (5.1)
 > DGsetup(L1);
 ${\mathrm{Lie algebra: Alg1}}$ (5.2)

Define a new basis for the Lie algebra Alg1.

 Alg1 > B := evalDG([e1 + e3, e4, e2 + e4, e3]);
 ${B}{≔}\left[{\mathrm{e1}}{+}{\mathrm{e3}}{,}{\mathrm{e4}}{,}{\mathrm{e2}}{+}{\mathrm{e4}}{,}{\mathrm{e3}}\right]$ (5.3)

Calculate the structure constants for the Lie algebra Alg1 with respect to the basis B.

 Alg1 > L2, M := LieAlgebraData(B, Alg2, "Matrix");
 ${\mathrm{L2}}{,}{M}{≔}\left[\left[{\mathrm{e1}}{,}{\mathrm{e2}}\right]{=}{2}{}{\mathrm{e1}}{-}{\mathrm{e2}}{+}{\mathrm{e3}}{-}{\mathrm{e4}}{,}\left[{\mathrm{e1}}{,}{\mathrm{e3}}\right]{=}{\mathrm{e1}}{-}{\mathrm{e2}}{+}{\mathrm{e3}}{,}\left[{\mathrm{e2}}{,}{\mathrm{e3}}\right]{=}{\mathrm{e2}}{-}{\mathrm{e3}}{,}\left[{\mathrm{e2}}{,}{\mathrm{e4}}\right]{=}{\mathrm{e2}}{-}{\mathrm{e3}}{-}{\mathrm{e4}}{,}\left[{\mathrm{e3}}{,}{\mathrm{e4}}\right]{=}{\mathrm{e1}}{+}{\mathrm{e2}}{-}{\mathrm{e3}}{-}{2}{}{\mathrm{e4}}\right]{,}\left[\begin{array}{rrrr}{1}& {0}& {0}& {0}\\ {0}& {0}& {1}& {0}\\ {1}& {0}& {0}& {1}\\ {0}& {1}& {1}& {0}\end{array}\right]$ (5.4)

The first item returned by LieAlgebraData is the structure equations for Alg1 with respect to the basis B.  This give us another Lie algebra which we have designated as Alg2.

 Alg1 > DGsetup(L2);
 ${\mathrm{Lie algebra: Alg2}}$ (5.5)

Of course, the Lie algebras Alg1 and Alg2 are isomorphic.  The second item returned by LieAlgebraData is a matrix defining an isomorphism from Alg2 to Alg1.  We can check this with the Query command.

 Alg2 > Query(Alg2, Alg1, M, "Homomorphism");
 ${\mathrm{true}}$ (5.6)

Construct a Lie algebra from a subalgebra of a Lie algebra

 Alg1 > with(DifferentialGeometry): with(LieAlgebras):

Define a Lie algebra algebra data structure and initialize the Lie algebra.

 > L1 := _DG([["LieAlgebra", Alg1, [4]], [[[2, 3, 1], 1], [[1, 4, 1], 2], [[2, 4, 2], 1], [[3, 4, 2], 1], [[3, 4, 3], 1]]]);
 ${\mathrm{L1}}{≔}\left[\left[{\mathrm{e1}}{,}{\mathrm{e4}}\right]{=}{2}{}{\mathrm{e1}}{,}\left[{\mathrm{e2}}{,}{\mathrm{e3}}\right]{=}{\mathrm{e1}}{,}\left[{\mathrm{e2}}{,}{\mathrm{e4}}\right]{=}{\mathrm{e2}}{,}\left[{\mathrm{e3}}{,}{\mathrm{e4}}\right]{=}{\mathrm{e2}}{+}{\mathrm{e3}}\right]$ (6.1)
 Alg1 > DGsetup(L1);
 ${\mathrm{Lie algebra: Alg1}}$ (6.2)

We check that the list of vectors S define a subalgebra of Alg1.

 Alg1 > S := [e1, e2, e3];
 ${S}{≔}\left[{\mathrm{e1}}{,}{\mathrm{e2}}{,}{\mathrm{e3}}\right]$ (6.3)
 Alg1 > Query(S, "Subalgebra");
 ${\mathrm{true}}$ (6.4)

We now create the Lie algebra data structure L4 which can then be used to initialize the subalgebra S as Lie algebra Alg4 in its own right.

 Alg1 > L4 := LieAlgebraData(S, Alg4);
 ${\mathrm{L4}}{≔}\left[\left[{\mathrm{e2}}{,}{\mathrm{e3}}\right]{=}{\mathrm{e1}}\right]$ (6.5)

Using the DiffererentialGeometry Library to create a Lie algebra

The DifferentialGeometry subpackage Library contains extensive tables of Lie algebras taken from the mathematics and mathematical physics literature.

The command References contain a list of the papers whose results are currently available in the Library.

 Alg1 > with(DifferentialGeometry): with(LieAlgebras): with(Library):
 > References(verbose);
 Gong, 1          Classification of Nilpotent Lie Algebras of Dimension 7( Over Algebraically Closed Fields and R)          PhD. Thesis,  University of Waterloo (1998)     Gonzalez-Lopez, 1          Lie algebras of vector fields in the real plane (with Kamran and Olver)          Proc. London Math Soc. Vol 64 (1992), 339--368     Kamke, 1           Differentialgleichungen           Chelsa Publ. Co. (1947)   Mubarakzyanov, 1          Lie algebras of dimmensions 3, 4          Izv. Vyssh. Uchebn. Zaved. Math 34(1963) 99.   Mubarakzyanov, 2          Lie algebras of dimension 5          Izv. Vyssh. Uchebn. Zaved. Math 34(1963) 99.   Mubarakzyanov, 3          Lie algebras of dimension 6          Izv. Vyssh. Uchebn. Zaved. Math 35(1963) 104.   Olver, 1:          Equivalence, Invariants and Symmetry, 472--473   Petrov, 1:          Einstein Spaces   Turkowski, 1:          Low dimensional real Lie algebras          JMP(29), 1990, 2139--2144   Turkowski, 2          Solvable Lie Algebras of dimension six          JMP(31), 1990, 1344--1350   Winternitz, 1:          Invariants of real low dimensional Lie algebras, (with Patera, Sharp and Zassenhaus)          JMP vol 17, No 6, June 1976, 966--994   $\left[\left[{"Gong"}{,}{1}\right]{,}\left[{"Gonzalez-Lopez"}{,}{1}\right]{,}\left[{"Kamke"}{,}{1}\right]{,}\left[{"Morozov"}{,}{1}\right]{,}\left[{"Mubarakyzanov"}{,}{1}\right]{,}\left[{"Mubarakyzanov"}{,}{2}\right]{,}\left[{"Mubarakyzanov"}{,}{3}\right]{,}\left[{"Olver"}{,}{1}\right]{,}\left[{"Petrov"}{,}{1}\right]{,}\left[{"Turkowski"}{,}{1}\right]{,}\left[{"Turkowski"}{,}{2}\right]{,}\left[{"USU"}{,}{2}\right]{,}\left[{"USU"}{,}{"2D"}\right]{,}\left[{"Winternitz"}{,}{1}\right]\right]$ (7.1)

The Browse command is used first to find the indices for each table in the Library and, second, to display the selected entries in the table.

 > L := Browse("Winternitz", 1);
 ${L}{≔}\left[\left[{3}{,}{0}\right]{,}\left[{3}{,}{1}\right]{,}\left[{3}{,}{2}\right]{,}\left[{3}{,}{3}\right]{,}\left[{3}{,}{4}\right]{,}\left[{3}{,}{5}\right]{,}\left[{3}{,}{6}\right]{,}\left[{3}{,}{7}\right]{,}\left[{3}{,}{8}\right]{,}\left[{3}{,}{9}\right]{,}\left[{4}{,}{0}\right]{,}\left[{4}{,}{1}\right]{,}\left[{4}{,}{2}\right]{,}\left[{4}{,}{3}\right]{,}\left[{4}{,}{4}\right]{,}\left[{4}{,}{5}\right]{,}\left[{4}{,}{6}\right]{,}\left[{4}{,}{7}\right]{,}\left[{4}{,}{8}\right]{,}\left[{4}{,}{9}\right]{,}\left[{4}{,}{10}\right]{,}\left[{4}{,}{11}\right]{,}\left[{4}{,}{12}\right]{,}\left[{5}{,}{0}\right]{,}\left[{5}{,}{1}\right]{,}\left[{5}{,}{2}\right]{,}\left[{5}{,}{3}\right]{,}\left[{5}{,}{4}\right]{,}\left[{5}{,}{5}\right]{,}\left[{5}{,}{6}\right]{,}\left[{5}{,}{7}\right]{,}\left[{5}{,}{8}\right]{,}\left[{5}{,}{9}\right]{,}\left[{5}{,}{10}\right]{,}\left[{5}{,}{11}\right]{,}\left[{5}{,}{12}\right]{,}\left[{5}{,}{13}\right]{,}\left[{5}{,}{14}\right]{,}\left[{5}{,}{15}\right]{,}\left[{5}{,}{16}\right]{,}\left[{5}{,}{17}\right]{,}\left[{5}{,}{18}\right]{,}\left[{5}{,}{19}\right]{,}\left[{5}{,}{20}\right]{,}\left[{5}{,}{21}\right]{,}\left[{5}{,}{22}\right]{,}\left[{5}{,}{23}\right]{,}\left[{5}{,}{24}\right]{,}\left[{5}{,}{25}\right]{,}\left[{5}{,}{26}\right]{,}\left[{5}{,}{27}\right]{,}\left[{5}{,}{28}\right]{,}\left[{5}{,}{29}\right]{,}\left[{5}{,}{30}\right]{,}\left[{5}{,}{31}\right]{,}\left[{5}{,}{32}\right]{,}\left[{5}{,}{33}\right]{,}\left[{5}{,}{34}\right]{,}\left[{5}{,}{35}\right]{,}\left[{5}{,}{36}\right]{,}\left[{5}{,}{37}\right]{,}\left[{5}{,}{38}\right]{,}\left[{5}{,}{39}\right]{,}\left[{5}{,}{40}\right]{,}\left[{6}{,}{1}\right]{,}\left[{6}{,}{2}\right]{,}\left[{6}{,}{3}\right]{,}\left[{6}{,}{4}\right]{,}\left[{6}{,}{5}\right]{,}\left[{6}{,}{6}\right]{,}\left[{6}{,}{7}\right]{,}\left[{6}{,}{8}\right]{,}\left[{6}{,}{9}\right]{,}\left[{6}{,}{10}\right]{,}\left[{6}{,}{11}\right]{,}\left[{6}{,}{12}\right]{,}\left[{6}{,}{13}\right]{,}\left[{6}{,}{14}\right]{,}\left[{6}{,}{15}\right]{,}\left[{6}{,}{16}\right]{,}\left[{6}{,}{17}\right]{,}\left[{6}{,}{18}\right]{,}\left[{6}{,}{19}\right]{,}\left[{6}{,}{20}\right]{,}\left[{6}{,}{21}\right]{,}\left[{6}{,}{22}\right]\right]$ (7.2)
 > Browse("Winternitz", 1, L[11 .. 15]);
 ${"Winternitz"}{,}{1}{,}\left[{4}{,}{0}\right]$ $\left[{}\right]$ ${\mathrm{___________________}}$ ${"Winternitz"}{,}{1}{,}\left[{4}{,}{1}\right]$ $\left[\left[{\mathrm{e2}}{,}{\mathrm{e4}}\right]{=}{\mathrm{e1}}{,}\left[{\mathrm{e3}}{,}{\mathrm{e4}}\right]{=}{\mathrm{e2}}\right]$ ${\mathrm{___________________}}$ ${"Winternitz"}{,}{1}{,}\left[{4}{,}{2}\right]$ $\left[\left[{\mathrm{e1}}{,}{\mathrm{e4}}\right]{=}{a}{}{\mathrm{e1}}{,}\left[{\mathrm{e2}}{,}{\mathrm{e4}}\right]{=}{\mathrm{e2}}{,}\left[{\mathrm{e3}}{,}{\mathrm{e4}}\right]{=}{\mathrm{e2}}{+}{\mathrm{e3}}\right]$ ${\mathrm{___________________}}$ ${"Winternitz"}{,}{1}{,}\left[{4}{,}{3}\right]$ $\left[\left[{\mathrm{e1}}{,}{\mathrm{e4}}\right]{=}{\mathrm{e1}}{,}\left[{\mathrm{e3}}{,}{\mathrm{e4}}\right]{=}{\mathrm{e2}}\right]$ ${\mathrm{___________________}}$ ${"Winternitz"}{,}{1}{,}\left[{4}{,}{4}\right]$ $\left[\left[{\mathrm{e1}}{,}{\mathrm{e4}}\right]{=}{\mathrm{e1}}{,}\left[{\mathrm{e2}}{,}{\mathrm{e4}}\right]{=}{\mathrm{e1}}{+}{\mathrm{e2}}{,}\left[{\mathrm{e3}}{,}{\mathrm{e4}}\right]{=}{\mathrm{e2}}{+}{\mathrm{e3}}\right]$ ${\mathrm{___________________}}$ (7.3)

The Retrieve command is then used to obtain the Lie algebra data structure for a given algebra.  The output of Retrieve is passed to the DGsetup command in order to initialize the Lie algebra.

 > L := Retrieve("Winternitz", 1, [4, 4], Alg1);
 ${L}{≔}\left[\left[{\mathrm{e1}}{,}{\mathrm{e4}}\right]{=}{\mathrm{e1}}{,}\left[{\mathrm{e2}}{,}{\mathrm{e4}}\right]{=}{\mathrm{e1}}{+}{\mathrm{e2}}{,}\left[{\mathrm{e3}}{,}{\mathrm{e4}}\right]{=}{\mathrm{e2}}{+}{\mathrm{e3}}\right]$ (7.4)
 > DGsetup(L);
 ${\mathrm{Lie algebra: Alg1}}$ (7.5)

DGsetup

The command DGsetup (in the DifferentialGeometry package) is used to store the structure equations for a Lie algebra in memory.  These structure equations can be accessed at any time during the Maple session.

The default labels for the basis elements of a Lie algebra are e1, e2, e3, ....  The default labels for the basis elements of the dual of the Lie algebra are theta1, theta2, theta3, ....

You may change these labels during execution of the DGsetup command by specifying either the the kernel letter or by specifying a full list of label names.  As always, these names must be unassigned Maple names.

We use the Lie algebra of 2x2 trace-free matrices (sl2) to illustrate these features of DGsetup.

 Alg1 > with(DifferentialGeometry): with(LieAlgebras):
 Alg1 > M := [Matrix([[1, 0], [0, -1]]), Matrix([[0, 1], [0, 0]]), Matrix([[0, 0], [1, 0]])];
 ${M}{≔}\left[\left[\begin{array}{rr}{1}& {0}\\ {0}& {-}{1}\end{array}\right]{,}\left[\begin{array}{rr}{0}& {1}\\ {0}& {0}\end{array}\right]{,}\left[\begin{array}{rr}{0}& {0}\\ {1}& {0}\end{array}\right]\right]$ (8.1)
 Alg1 > L := LieAlgebraData(M, Alg1);
 ${L}{≔}\left[\left[{\mathrm{e1}}{,}{\mathrm{e2}}\right]{=}{2}{}{\mathrm{e2}}{,}\left[{\mathrm{e1}}{,}{\mathrm{e3}}\right]{=}{-}{2}{}{\mathrm{e3}}{,}\left[{\mathrm{e2}}{,}{\mathrm{e3}}\right]{=}{\mathrm{e1}}\right]$ (8.2)

Example 1.  Initialize the Lie algebra Alg1 and label the basis elements E1, E2, E3, epsilon1, epsilon2, epsilon3.

 Alg1 > DGsetup(L, [E], [epsilon]);
 ${\mathrm{Lie algebra: Alg1}}$ (8.3)
 Alg1 > MultiplicationTable("LieTable");
 $\left[\begin{array}{ccccc}{}& {\mathrm{|}}& {\mathrm{E1}}& {\mathrm{E2}}& {\mathrm{E3}}\\ {}& {\mathrm{---}}& {\mathrm{----}}& {\mathrm{----}}& {\mathrm{----}}\\ {\mathrm{E1}}& {\mathrm{|}}& {0}& {2}{}{\mathrm{E2}}& {-}{2}{}{\mathrm{E3}}\\ {\mathrm{E2}}& {\mathrm{|}}& {-}{2}{}{\mathrm{E2}}& {0}& {\mathrm{E1}}\\ {\mathrm{E3}}& {\mathrm{|}}& {2}{}{\mathrm{E3}}& {-}{\mathrm{E1}}& {0}\end{array}\right]$ (8.4)
 Alg1 > MultiplicationTable("ExteriorDerivative");
 ${d}{}\left({\mathrm{ε1}}\right){=}{-}{\mathrm{ε2}}{}{\mathrm{^}}{}{\mathrm{ε3}}$ ${d}{}\left({\mathrm{ε2}}\right){=}{-}{2}{}{\mathrm{ε1}}{}{\mathrm{^}}{}{\mathrm{ε2}}$ ${d}{}\left({\mathrm{ε3}}\right){=}{2}{}{\mathrm{ε1}}{}{\mathrm{^}}{}{\mathrm{ε3}}$ (8.5)

Example 2.  Initialize the Lie algebra Alg1 and label the basis elements X, Y, Z, alpha, beta, sigma.

 Alg1 > DGsetup(L, [X, Y, Z], [alpha, beta, sigma]);
 ${\mathrm{Lie algebra: Alg1}}$ (8.6)
 Alg1 > MultiplicationTable("LieTable");
 $\left[\begin{array}{ccccc}{}& {\mathrm{|}}& {X}& {Y}& {Z}\\ {}& {\mathrm{---}}& {\mathrm{----}}& {\mathrm{----}}& {\mathrm{----}}\\ {X}& {\mathrm{|}}& {0}& {2}{}{Y}& {-}{2}{}{Z}\\ {Y}& {\mathrm{|}}& {-}{2}{}{Y}& {0}& {X}\\ {Z}& {\mathrm{|}}& {2}{}{Z}& {-}{X}& {0}\end{array}\right]$ (8.7)
 Alg1 > MultiplicationTable("ExteriorDerivative");
 ${d}{}\left({\mathrm{α}}\right){=}{-}{\mathrm{β}}{}{\mathrm{^}}{}{\mathrm{σ}}$ ${d}{}\left({\mathrm{β}}\right){=}{-}{2}{}{\mathrm{α}}{}{\mathrm{^}}{}{\mathrm{β}}$ ${d}{}\left({\mathrm{σ}}\right){=}{2}{}{\mathrm{α}}{}{\mathrm{^}}{}{\mathrm{σ}}$ (8.8)

Exercises

Exercise 1

Initialize the 5 dimensional Lie algebra Lie1 with structure equations [e2, e4] = e1, [e3, e5] = e1.  Display the multiplication table.

 Alg1 > with(DifferentialGeometry): with(LieAlgebras):

Solution

Remember that unassigned names must be used here.

 Alg1 > Eq := [[x2, x4] = x1, [x3, x5] = x1];
 ${\mathrm{Eq}}{≔}\left[\left[{\mathrm{x2}}{,}{\mathrm{x4}}\right]{=}{\mathrm{x1}}{,}\left[{\mathrm{x3}}{,}{\mathrm{x5}}\right]{=}{\mathrm{x1}}\right]$ (9.1.1.1)
 Alg1 > L := LieAlgebraData(Eq, [x1, x2, x3, x4, x5], Lie1);
 ${L}{≔}\left[\left[{\mathrm{e2}}{,}{\mathrm{e4}}\right]{=}{\mathrm{e1}}{,}\left[{\mathrm{e3}}{,}{\mathrm{e5}}\right]{=}{\mathrm{e1}}\right]$ (9.1.1.2)
 Alg1 > DGsetup(L);
 ${\mathrm{Lie algebra: Lie1}}$ (9.1.1.3)
 Lie1 > MultiplicationTable("LieTable");
 $\left[\begin{array}{ccccccc}{}& {\mathrm{|}}& {\mathrm{e1}}& {\mathrm{e2}}& {\mathrm{e3}}& {\mathrm{e4}}& {\mathrm{e5}}\\ {}& {\mathrm{---}}& {\mathrm{----}}& {\mathrm{----}}& {\mathrm{----}}& {\mathrm{----}}& {\mathrm{----}}\\ {\mathrm{e1}}& {\mathrm{|}}& {0}& {0}& {0}& {0}& {0}\\ {\mathrm{e2}}& {\mathrm{|}}& {0}& {0}& {0}& {\mathrm{e1}}& {0}\\ {\mathrm{e3}}& {\mathrm{|}}& {0}& {0}& {0}& {0}& {\mathrm{e1}}\\ {\mathrm{e4}}& {\mathrm{|}}& {0}& {-}{\mathrm{e1}}& {0}& {0}& {0}\\ {\mathrm{e5}}& {\mathrm{|}}& {0}& {0}& {-}{\mathrm{e1}}& {0}& {0}\end{array}\right]$ (9.1.1.4)

Exercise 2

Initialize the 3 dimensional Lie algebra of 2x2 upper triangular matrices.  Call the algebra Upper.

Use the Query command to verify that this Lie algebra is solvable.

 Lie1 > with(DifferentialGeometry): with(LieAlgebras):

Solution

 > M := map(Matrix, [[[1, 0], [0, 0]], [[0, 1], [0, 0]], [[0, 0], [0, 1]]]);
 ${M}{≔}\left[\left[\begin{array}{rr}{1}& {0}\\ {0}& {0}\end{array}\right]{,}\left[\begin{array}{rr}{0}& {1}\\ {0}& {0}\end{array}\right]{,}\left[\begin{array}{rr}{0}& {0}\\ {0}& {1}\end{array}\right]\right]$ (9.2.1.1)
 Lie1 > L := LieAlgebraData(M, Upper);
 ${L}{≔}\left[\left[{\mathrm{e1}}{,}{\mathrm{e2}}\right]{=}{\mathrm{e2}}{,}\left[{\mathrm{e2}}{,}{\mathrm{e3}}\right]{=}{\mathrm{e2}}\right]$ (9.2.1.2)
 Lie1 > DGsetup(L);
 ${\mathrm{Lie algebra: Upper}}$ (9.2.1.3)
 Upper > Query("Solvable");
 ${\mathrm{true}}$ (9.2.1.4)

Exercise 3

[i]  Define the 6 vector fields in Euclidean 3 space E3 (with coordinates [x, y, z]) whose flows are translations in the x, y, and z directions and rotations about the x, y, and z axes.  Call the Lie algebra euc.  Check that these vector fields form a Lie algebra.

[ii]  Initialize this Lie algebra and use a basis T1, T2, T3, R1, R2, R3, where the T's are translations and the R's are rotations.

[iii]  Use the LeviDecomposition command to show that the radical of euc is [T1, T2, T3] and that a semisimple complement to the radical is [R1, R2, R3].

 Upper > with(DifferentialGeometry): with(LieAlgebras):

Solution

Part [i]

 Upper > DGsetup([x, y, z], E3):
 E3 > Gamma := evalDG([D_x, D_y, D_z, y*D_z - z*D_y, x*D_z - z*D_x, x*D_y - y*D_x]);
 ${\mathrm{Γ}}{≔}\left[{\mathrm{D_x}}{,}{\mathrm{D_y}}{,}{\mathrm{D_z}}{,}{-}{z}{}{\mathrm{D_y}}{+}{y}{}{\mathrm{D_z}}{,}{-}{z}{}{\mathrm{D_x}}{+}{x}{}{\mathrm{D_z}}{,}{-}{y}{}{\mathrm{D_x}}{+}{x}{}{\mathrm{D_y}}\right]$ (9.3.1.1)
 E3 > L := LieAlgebraData(Gamma, euc3);
 ${L}{≔}\left[\left[{\mathrm{e1}}{,}{\mathrm{e5}}\right]{=}{\mathrm{e3}}{,}\left[{\mathrm{e1}}{,}{\mathrm{e6}}\right]{=}{\mathrm{e2}}{,}\left[{\mathrm{e2}}{,}{\mathrm{e4}}\right]{=}{\mathrm{e3}}{,}\left[{\mathrm{e2}}{,}{\mathrm{e6}}\right]{=}{-}{\mathrm{e1}}{,}\left[{\mathrm{e3}}{,}{\mathrm{e4}}\right]{=}{-}{\mathrm{e2}}{,}\left[{\mathrm{e3}}{,}{\mathrm{e5}}\right]{=}{-}{\mathrm{e1}}{,}\left[{\mathrm{e4}}{,}{\mathrm{e5}}\right]{=}{\mathrm{e6}}{,}\left[{\mathrm{e4}}{,}{\mathrm{e6}}\right]{=}{-}{\mathrm{e5}}{,}\left[{\mathrm{e5}}{,}{\mathrm{e6}}\right]{=}{\mathrm{e4}}\right]$ (9.3.1.2)

[If the vector fields Gamma do not form a Lie algebra, the empty list is returned.]

Part [ii]

 E3 > DGsetup(L, [T1, T2, T3, R1, R2, R3], [theta]);
 ${\mathrm{Lie algebra: euc3}}$ (9.3.1.3)
 euc3 > MultiplicationTable("LieTable");
 $\left[\begin{array}{cccccccc}{}& {\mathrm{|}}& {\mathrm{T1}}& {\mathrm{T2}}& {\mathrm{T3}}& {\mathrm{R1}}& {\mathrm{R2}}& {\mathrm{R3}}\\ {}& {\mathrm{---}}& {\mathrm{----}}& {\mathrm{----}}& {\mathrm{----}}& {\mathrm{----}}& {\mathrm{----}}& {\mathrm{----}}\\ {\mathrm{T1}}& {\mathrm{|}}& {0}& {0}& {0}& {0}& {\mathrm{T3}}& {\mathrm{T2}}\\ {\mathrm{T2}}& {\mathrm{|}}& {0}& {0}& {0}& {\mathrm{T3}}& {0}& {-}{\mathrm{T1}}\\ {\mathrm{T3}}& {\mathrm{|}}& {0}& {0}& {0}& {-}{\mathrm{T2}}& {-}{\mathrm{T1}}& {0}\\ {\mathrm{R1}}& {\mathrm{|}}& {0}& {-}{\mathrm{T3}}& {\mathrm{T2}}& {0}& {\mathrm{R3}}& {-}{\mathrm{R2}}\\ {\mathrm{R2}}& {\mathrm{|}}& {-}{\mathrm{T3}}& {0}& {\mathrm{T1}}& {-}{\mathrm{R3}}& {0}& {\mathrm{R1}}\\ {\mathrm{R3}}& {\mathrm{|}}& {-}{\mathrm{T2}}& {\mathrm{T1}}& {0}& {\mathrm{R2}}& {-}{\mathrm{R1}}& {0}\end{array}\right]$ (9.3.1.4)

Part [iii]  The LeviDecomposition procedure returns a list with two lists of vectors.  The first list is a basis for the radical and the second list is a semisimple complement to the radical.

 euc3 > LD := LeviDecomposition();
 ${\mathrm{LD}}{≔}\left[\left[{\mathrm{T1}}{,}{\mathrm{T2}}{,}{\mathrm{T3}}\right]{,}\left[{\mathrm{R1}}{,}{\mathrm{R2}}{,}{\mathrm{R3}}\right]\right]$ (9.3.1.5)

Exercise 4

[i]  The paper Solvable Lie Algebras of dimension six by P. Turkowski (JMP(31), 1990) contains a complete list of all 6 dimensional Lie algebras with nilradicals of dimensions 3, 4, and 5.  Retrieve the the first and last Lie algebras in Turkwoski's classification and initialize.

[ii]  Use the command Nilradical to find the nilradical of these two Lie algebras.

 euc3 > with(DifferentialGeometry): with(LieAlgebras): with(Library):

Solution

We use the References command to find that the paper in question is referred to in the DifferentialGeometry library as "Turkowski", 2.

 > References(verbose);
 Gong, 1          Classification of Nilpotent Lie Algebras of Dimension 7( Over Algebraically Closed Fields and R)          PhD. Thesis,  University of Waterloo (1998)     Gonzalez-Lopez, 1          Lie algebras of vector fields in the real plane (with Kamran and Olver)          Proc. London Math Soc. Vol 64 (1992), 339--368     Kamke, 1           Differentialgleichungen           Chelsa Publ. Co. (1947)   Mubarakzyanov, 1          Lie algebras of dimmensions 3, 4          Izv. Vyssh. Uchebn. Zaved. Math 34(1963) 99.   Mubarakzyanov, 2          Lie algebras of dimension 5          Izv. Vyssh. Uchebn. Zaved. Math 34(1963) 99.   Mubarakzyanov, 3          Lie algebras of dimension 6          Izv. Vyssh. Uchebn. Zaved. Math 35(1963) 104.   Olver, 1:          Equivalence, Invariants and Symmetry, 472--473   Petrov, 1:          Einstein Spaces   Turkowski, 1:          Low dimensional real Lie algebras          JMP(29), 1990, 2139--2144   Turkowski, 2          Solvable Lie Algebras of dimension six          JMP(31), 1990, 1344--1350   Winternitz, 1:          Invariants of real low dimensional Lie algebras, (with Patera, Sharp and Zassenhaus)          JMP vol 17, No 6, June 1976, 966--994   $\left[\left[{"Gong"}{,}{1}\right]{,}\left[{"Gonzalez-Lopez"}{,}{1}\right]{,}\left[{"Kamke"}{,}{1}\right]{,}\left[{"Morozov"}{,}{1}\right]{,}\left[{"Mubarakyzanov"}{,}{1}\right]{,}\left[{"Mubarakyzanov"}{,}{2}\right]{,}\left[{"Mubarakyzanov"}{,}{3}\right]{,}\left[{"Olver"}{,}{1}\right]{,}\left[{"Petrov"}{,}{1}\right]{,}\left[{"Turkowski"}{,}{1}\right]{,}\left[{"Turkowski"}{,}{2}\right]{,}\left[{"USU"}{,}{2}\right]{,}\left[{"USU"}{,}{"2D"}\right]{,}\left[{"Winternitz"}{,}{1}\right]\right]$ (9.4.1.1)

Next, we find the indices for the "Turkowski", 2 table of Lie algebras.

 euc3 > Indices := Browse("Turkowski", 2);
 ${\mathrm{Indices}}{≔}\left[\left[{6}{,}{1}\right]{,}\left[{6}{,}{2}\right]{,}\left[{6}{,}{3}\right]{,}\left[{6}{,}{4}\right]{,}\left[{6}{,}{5}\right]{,}\left[{6}{,}{6}\right]{,}\left[{6}{,}{7}\right]{,}\left[{6}{,}{8}\right]{,}\left[{6}{,}{9}\right]{,}\left[{6}{,}{10}\right]{,}\left[{6}{,}{11}\right]{,}\left[{6}{,}{12}\right]{,}\left[{6}{,}{13}\right]{,}\left[{6}{,}{14}\right]{,}\left[{6}{,}{15}\right]{,}\left[{6}{,}{16}\right]{,}\left[{6}{,}{17}\right]{,}\left[{6}{,}{18}\right]{,}\left[{6}{,}{19}\right]{,}\left[{6}{,}{20}\right]{,}\left[{6}{,}{21}\right]{,}\left[{6}{,}{22}\right]{,}\left[{6}{,}{23}\right]{,}\left[{6}{,}{24}\right]{,}\left[{6}{,}{25}\right]{,}\left[{6}{,}{26}\right]{,}\left[{6}{,}{27}\right]{,}\left[{6}{,}{28}\right]{,}\left[{6}{,}{29}\right]{,}\left[{6}{,}{30}\right]{,}\left[{6}{,}{31}\right]{,}\left[{6}{,}{32}\right]{,}\left[{6}{,}{33}\right]{,}\left[{6}{,}{34}\right]{,}\left[{6}{,}{35}\right]{,}\left[{6}{,}{36}\right]{,}\left[{6}{,}{37}\right]{,}\left[{6}{,}{38}\right]{,}\left[{6}{,}{39}\right]{,}\left[{6}{,}{40}\right]\right]$ (9.4.1.2)

Retrieve the first and last Lie algebras in the table.  The second and third elements in L1 and L2 given the labels for the basis of the Lie algebras to match the notation found in the paper.

 euc3 > L1 := Retrieve("Turkowski", 2, [6, 1], Turk1);
 ${\mathrm{L1}}{≔}\left[\left[{\mathrm{e1}}{,}{\mathrm{e3}}\right]{=}{\mathrm{α}}{}{\mathrm{e3}}{,}\left[{\mathrm{e1}}{,}{\mathrm{e4}}\right]{=}{\mathrm{γ}}{}{\mathrm{e4}}{,}\left[{\mathrm{e1}}{,}{\mathrm{e6}}\right]{=}{\mathrm{e6}}{,}\left[{\mathrm{e2}}{,}{\mathrm{e3}}\right]{=}{\mathrm{β}}{}{\mathrm{e3}}{,}\left[{\mathrm{e2}}{,}{\mathrm{e4}}\right]{=}{\mathrm{δ}}{}{\mathrm{e4}}{,}\left[{\mathrm{e2}}{,}{\mathrm{e5}}\right]{=}{\mathrm{e5}}\right]{,}\left[{\mathrm{x1}}{,}{\mathrm{x2}}{,}{\mathrm{n1}}{,}{\mathrm{n2}}{,}{\mathrm{n3}}{,}{\mathrm{n4}}\right]{,}\left[{\mathrm{θ}}\right]$ (9.4.1.3)
 euc3 > L2 := Retrieve("Turkowski", 2, [6, 40], Turk40);
 ${\mathrm{L2}}{≔}\left[\left[{\mathrm{e1}}{,}{\mathrm{e2}}\right]{=}{-}{\mathrm{e3}}{,}\left[{\mathrm{e1}}{,}{\mathrm{e4}}\right]{=}{\mathrm{e5}}{,}\left[{\mathrm{e1}}{,}{\mathrm{e5}}\right]{=}{-}{\mathrm{e4}}{,}\left[{\mathrm{e2}}{,}{\mathrm{e6}}\right]{=}{\mathrm{e6}}{,}\left[{\mathrm{e4}}{,}{\mathrm{e5}}\right]{=}{-}{\mathrm{e3}}\right]{,}\left[{\mathrm{x1}}{,}{\mathrm{x2}}{,}{\mathrm{n1}}{,}{\mathrm{n2}}{,}{\mathrm{n3}}{,}{\mathrm{n4}}\right]{,}\left[{\mathrm{θ}}\right]$ (9.4.1.4)

Initialize each Lie algebra and compute the nilradicals.

 euc3 > DGsetup(L1);
 ${\mathrm{Lie algebra: Turk1}}$ (9.4.1.5)
 $\left[{\mathrm{n1}}{,}{\mathrm{n2}}{,}{\mathrm{n3}}{,}{\mathrm{n4}}\right]$ (9.4.1.6)
 ${\mathrm{Lie algebra: Turk40}}$ (9.4.1.7)
 $\left[{\mathrm{n1}}{,}{\mathrm{n2}}{,}{\mathrm{n3}}{,}{\mathrm{n4}}\right]$ (9.4.1.8)